POGLAVJE II

II.6 Polinomi in trigonometrične funkcije kot interpolacijske funkcije

1. Izbira polinomskih in trigonometričnih funkcij
2. Izračun posplošene mase, posplošene togosti in posplošene geometrijske togosti
3. Izračun lastne frekvence ter kritične uklonske sile

1. Izbira polinomskih in trigonometričnih funkcij

Prejšnji podpoglavji sta pokazali, da je analitično reševanjem diferencialne enačbe pomika lastnega prečnega nihanja zelo omejeno celo pri enostavnih konstrukcijah. Zato se pri izračunu približka lastne frekvence namesto matematično popolnoma natančne rešitve uporabljajo smiselno izbrane funkcije (ki seveda ne ustrezajo popolnoma diferencialni enačbi oz. njeni rešitvi, vendar zaradi svoje relativne enostavnosti omogočajo izračun dovolj kvalitetnih približkov prečnih pomikov in posledično lastnih frekvenc). Kot takšne zelo primerne interpolacijske funkcije se izkažejo polinomi, v nekaterih situacijah pa tudi trigonometrične funkcije.
V tem poglavju bo analizirana konzola iz vaje 10 (http://www.geocities.com/mcsDISK/e10.pdf) in zato se tudi datoteka opremi z tem naslovom. Ker pa bo prikazanih več različnih interpolacijskih funkcij, se doda še podnaslov.
Kot prva se izbere najpreprostejša še uporabna polinomska interpolacijska funkcija – popolni polinom drugega reda oz. kvadratni polinom (linearna funkcija pomikov namreč ni uporabna, saj je zmožna opisati pomike zgolj kot pomike togega telesa).
Splošni polinom se lahko zapiše kar na način, ki je uporabniku najpreprostejši, torej z zapisom vsake konstante posebej (v nadaljevanju pa bo prikazan tudi elegantnejši zapis splošnega polinoma):

Vaja 10

■ Prva funkcija

```
In[1]:= ψ1[x_] := a1 + a2 x + a3 x^2 ;
```

Ker v funkciji nastopajo tri neznane konstante, se lahko upoštevajo zgolj trije kinematični pogoji (dva na začetku, torej na mestu vpetja, kjer je tudi izhodišče koordinatnega sistema, ter neničelni pogoj (ki zagotavlja neničelno rešitev), npr. pomik na prostem koncu konzole (namesto te najbolj očitne in logične lokacije bi se lahko uporabila tudi kakšna druga, razen na mestu vpetja, kjer bi se definiral nek, od nič različen zasuk):

```
In[2]:= resitve = Solve[{ψ1[0] == 0, ψ1'[0] == 0, ψ1[L] == 1},
          {a1, a2, a3}]
```

Out[2]= $\left\{\left\{a3 \to \frac{1}{L^2},\ a1 \to 0,\ a2 \to 0\right\}\right\}$

Izračunane vrednosti je potrebno še dodeliti še vedno neznanim konstantam:

```
In[3]:= a1=a1/.resitve[[1]];
        a2=a2/.resitve[[1]];
        a3=a3/.resitve[[1]];
```

kar vodi do zelo enostavne funkcije pomika:

In[6]:= $\psi 1[x]$

Out[6]= $\frac{x^2}{L^2}$

Zaradi kasnejše lažje grafične predstavitve različnih interpolacijskih funkcij se sedaj pridobljena funkcija izriše, npr. v svetlo modri barvi (z opcijo *PlotStyle->RGBColor[0,1,1])*, slika pa se shrani v ustrezno spremenljivko, npr. *funkcija1*:

```
In[7]:= funkcija1 = Plot[ψ1[x] /. L -> 1, {x, 0, 1},
          PlotStyle -> RGBColor[0, 1, 1]];
```



2. Izračun posplošene mase, posplošene togosti in posplošene geometrijske togosti

Za izbrano funkcijo se sedaj lahko izračunajo posplošena masa:

In[8]:= $mz = \int_0^L m\, \psi 1[x]^2\, dx$

Out[8]= $\frac{L\, m}{5}$

posplošena togost

In[9]:= $kz = \int_0^L EI\, \psi 1''[x]^2\, dx$

Out[9]= $\frac{4\, EI}{L^3}$

ter posplošena geometrijska togost zaradi koncentrirane vertikalne sile P_{ukl} na vrhu konstrukcije, ki povzroča konstantno tlačno osno silo P_{ukl} (vzdolž celotne konstrukcije):

In[10]:= $kg = \int_0^L Pukl\, \psi 1'[x]^2\, dx$

Out[10]= $\frac{4\, Pukl}{3\, L}$

3. Izračun lastne frekvence ter kritične uklonske sile

S pomočjo pridobljenih vrednosti (zapisanih sicer v obliki analitičnih izrazov) je mogoče izračunati približek kritične uklonske sile (ker dejansko ne gre za rešitev diferencialne enačbe, bo ta vrednost tako zgolj približek rešitve, dobljene z reševanjem diferencialne enačbe):

In[11]:= `N[Solve[kz - kg == 0, Pukl]]`

Out[11]= $\left\{\left\{Pukl \to \frac{3.\, EI}{L^2}\right\}\right\}$

ter približek prve lastne krožne frekvence (namesto znaka za koren iz menije se kar lahko uporabi ukaz *Sqrt[argument]*:

In[12]:= $N\left[Sqrt\left[\frac{kz}{mz}\right]\right]$

Out[12]= $4.47214 \sqrt{\frac{EI}{L^4\, m}}$

Izračuna pa se lahko še posplošena obtežba, potrebna za časovni izračun odziva z diferencialno enačbo po času:

In[13]:= $pz = \int_0^L \psi 1[x]\, dx$

Out[13]= $\frac{L}{3}$

Ker za splošno konstrukcijo analitični izrazi za lastno frekvenco in kritično uklonsko silo običajno niso znani (kot so slučajno sedaj), je mogoče o kvaliteti dobljenih dveh izrazov soditi le na osnovi primerjave rezultatov z drugo interpolacijsko funkcijo ali morebitnim izračunom z drugačnim računskim modelom (npr. z metodo končnih elementov).

Ker je očitno, da je mogoče poleg že uporabljenih robnih pogojev upoštevati še nekatere druge (npr., da je upogibni moment na prostem koncu konzole enak nič), se zato ponovno poišče nova interpolacijska funkcija, npr. spet v obliki polinoma.
Da se ohrani tudi prva funkcija, se kot neznane konstante izberejo npr. b_1, b_2,....

■ Druga funkcija

```
In[14]:= ψ2[x_] := b1 + b2 x + b3 x^2 + b4 x^3;
         resitve = Solve[{ψ2[0] == 0, ψ2'[0] == 0, ψ2[L] == 1, ψ2''[L] == 0}
           , {b1, b2, b3, b4}];
         b1 = b1 /. resitve[[1]];
         b2 = b2 /. resitve[[1]];
         b3 = b3 /. resitve[[1]];
         b4 = b4 /. resitve[[1]];
```

Funkcija je tako:

```
In[20]:= ψ2[x]
```

Out[20]= $\frac{3\,x^2}{2\,L^2} - \frac{x^3}{2\,L^3}$

oziroma zapisana v bolj kompaktni obliki:

```
In[21]:= Simplify[%]
```

Out[21]= $\frac{(3\,L - x)\,x^2}{2\,L^3}$

Za boljšo predstavo o novi funkciji si je primerno še izrisati njeno sliko (npr. v svetlo vijolični barvi):

```
In[22]:= funkcija2 = Plot[ψ2[x] /. L -> 1, {x, 0, 1},
           PlotStyle -> RGBColor[1, 0, 1]];
```



ki pa, vsaj na prvi pogled, ne izkazuje nobenega bistvenega odstopanja glede na prvo funkcijo in zato se izračunajo še vsi pripadajoči posplošeni dinamični parametri, ki nudijo dodaten, bolj kritičen, natančen in objektiven vpogled v uspešnost oz. neuspešnost izbrane funkcije:

```
In[23]:= mz = ∫_0^L m ψ2[x]^2 dx
         kz = ∫_0^L EI ψ2''[x]^2 dx
         kg = ∫_0^L Pukl ψ2'[x]^2 dx

Out[23]= 33 L m / 140

Out[24]= 3 EI / L^3

Out[25]= 6 Pukl / (5 L)
```

Nato se lahko izračunajo še kritična uklonska sila, približek prve lastne frekvence ter posplošena obtežba:

```
In[26]:= N[Solve[kz - kg == 0, Pukl]]
         N[Sqrt[kz/mz]]
         pz = ∫_0^L ψ2[x] dx

Out[26]= {{Pukl → 2.5 EI / L^2}}

Out[27]= 3.56753 Sqrt[EI / (L^4 m)]

Out[28]= 3 L / 8
```

Iz primerjave rezultatov obeh funkcij je očitno, da je sedaj dobljeni približek prve lastne frekvence očitno boljši (saj je manjši), kar jasno potrjuje tudi primerjava z (slučajno) znanim analitičnim izrazom iz literature $\omega = 3.516 \cdot \sqrt{\frac{EI}{m \cdot L^4}}$ oz. numerično pridobljeno vrednostjo $\omega = 3.51602 \cdot \sqrt{\frac{EI}{m \cdot L^4}}$ iz podpoglavja II.5.

Boljši je tudi približek kritične uklonske sile (korektna vrednost namreč znaša $N_{cr} = P_{ukl} = \frac{\pi^2 \cdot EI}{4 \cdot L^2} = \frac{2.467 \cdot EI}{L^2}$), kar je posledica dejstva, da je bila izbrana kvalitetnejša interpolacijska funkcija.

Brez primerjave rezultatov z znanima analitičnima izrazoma bi se na osnovi primerjave vrednosti obeh funkcij zgolj zaznal premik rezultatov k točnim vrednostim, zaradi relativno velikih razlik med rezultati obeh funkcij (17 % za frekvenco oz. 20 % za kritično uklonsko silo) pa še vedno pa ne bi bilo mogoče oceniti uspešnosti druge interpolacijske funkcije.

Ker pa je na razpolago še ena informacija, ki velja med nihanjem (prečna sila na prostem koncu je enaka nič), se lahko rezultat še poskuša izboljšati z novo interpolacijsko funkcijo, ki bo sicer spet polinom, ampak višjega reda.

Čeprav je mogoče definirati polinom s tudi pristopom, uporabljenim pri prvih dveh funkcijah, bo sedaj iz demonstracijskih razlogov polinom kreiran drugače.

Tako se najprej kreirata spremenljivka *polinom*, v katero bodo shranjeni členi polinoma, ter spremenljivka *neznanke*, torej niz (vektor), v katerega se v zanki zapišejo vse neznanke – konstante, označene npr. z K[0], K[1], K[2],...:

■ Tretja funkcija

```
In[29]:= polinom = 0;
         neznanke = Table[K[i], {i, 0, 4}];
```

Sedaj se v spremenljivko *polinom* (ki mora že obvezno imeti začetno vrednost, torej mora biti definirana) v zanki dodajo vsi členi polinoma, kar se kompaktno (v načinu sintakse jezika C++) zapiše kot *polinom+= K[i] x^i*, kar je enako kot *polinom= polinom + K[i] x^i*:

```
In[31]:= Do[polinom += K[i] x^i, {i, 0, 4}]
```

Sedaj je potrebno še definirati tretjo interpolacijsko funkcijo kot:

```
In[32]:= ψ3[x_] = polinom
```

Out[32]= $K[0] + x\,K[1] + x^2\,K[2] + x^3\,K[3] + x^4\,K[4]$

Neznane konstante se spet izračunajo s pomočjo rešitve sistema linearnih enačb:

```
In[33]:= resitve = Solve[{ψ3[0] == 0, ψ3'[0] == 0, ψ3[L] == 1, ψ3''[L] == 0,
            ψ3'''[L] == 0}, neznanke]
```

Out[33]= $\left\{\left\{K[2] \to \frac{2}{L^2},\ K[0] \to 0,\ K[1] \to 0,\ K[3] \to -\frac{4}{3\,L^3},\ K[4] \to \frac{1}{3\,L^4}\right\}\right\}$

Izračunane vrednosti je potrebno še dodeliti konstantam, kar se sedaj lahko izvede zelo elegantno kar v zanki:

```
In[34]:= Do[K[i] = K[i] /. resitve[[1]], {i, 0, 4}];
```

Nova interpolacijska funkcija je tako:

```
In[35]:= ψ3[x]
```

Out[35]= $\frac{2\,x^2}{L^2} - \frac{4\,x^3}{3\,L^3} + \frac{x^4}{3\,L^4}$

Njena slika je spet podobna slikama prejšnjih dveh funkcij (ker barva izrisa ni posebej podana, bo izrisana v osnovni, črni barvi):

```
In[36]:= funkcija3 = Plot[ψ3[x] /. L -> 1, {x, 0, 1}];
```



Ker ni opaziti nobene očitne anomalije pridobljene funkcije, se zato izračunajo še pripadajoči posplošeni dinamični parametri, ki so:

In[37]:= $mz = \int_0^L m\, \psi 3[x]^2\, dx$

$kz = \int_0^L EI\, \psi 3''[x]^2\, dx$

$kg = \int_0^L Pukl\, \psi 3'[x]^2\, dx$

Out[37]= $\frac{104\, L\, m}{405}$

Out[38]= $\frac{16\, EI}{5\, L^3}$

Out[39]= $\frac{8\, Pukl}{7\, L}$

Kritična uklonska sila, približek prve lastne frekvence ter posplošena obtežba, ki pa predstavljajo pravo merilo uspešnosti nove funkcije, pa so:

```
In[40]:= N[Solve[kz - kg == 0, Pukl]]
         N[Sqrt[kz/mz]]
         pz = ∫_0^L ψ3[x] dx
```

$$\text{Out[40]= } \left\{\left\{\text{Pukl} \to \frac{2.8\,\text{EI}}{\text{L}^2}\right\}\right\}$$

$$\text{Out[41]= } 3.53009\sqrt{\frac{\text{EI}}{\text{L}^4\,\text{m}}}$$

$$\text{Out[42]= } \frac{2\,\text{L}}{5}$$

Iz primerjave rezultatov vseh funkcij je očitno, da je sedaj dobljeni približek prve lastne frekvence očitno še boljši (saj je manjši, ampak zgolj samo še za 1 % glede na prejšnji rezultat), kar potrjuje tudi primerjava z znanim analitičnim izrazom (ki pa seveda v splošnem ni znan).

Nasprotno pa je približek kritične uklonske sile nekoliko slabši (12 %, a še vedno boljši kot pri prvi funkciji), kar je posledica dejstva, da prečna sila na koncu konzole med uklonom ni enaka nič.

To tako potrjuje znano dejstvo, da sta funkciji prečnega pomika za račun dinamičnega odziva (lastne frekvence) in uklonskih problemov dejansko različni, saj sta tudi dobljeni kot rešitvi dveh različnih diferencialnih enačb.

Če se želi poiskati kvalitetnejša funkcija za račun uklona, je potrebno upoštevati, da zaradi izključno vertikalne obtežbe pri uklonu ne nastopi horizontalna reakcija in je zato posledično prečna sila torej enaka nič na začetku konzole (vpetem delu), kar se upošteva kot robni pogoj pri četrti interpolacijski funkciji:

Spet se definira vektor neznanih konstant, tokrat npr. M[0], M[1], M[2],..., da se ohranijo definirane vrednosti konstant K[0], K[1], K[2],....

■ Cetrta funkcija - posebej za iskanje kriticne uklonske sile

```
In[43]:= polinom = 0;
         neznanke = Table[M[i], {i, 0, 4}];
         Do[polinom += M[i] x^i, {i, 0, 4}]
         ψ4[x_] = polinom
```

$$\text{Out[46]= } \text{M}[0] + \text{x}\,\text{M}[1] + \text{x}^2\,\text{M}[2] + \text{x}^3\,\text{M}[3] + \text{x}^4\,\text{M}[4]$$

Na osnovi robnih pogojev se poiščejo še neznane konstante:

```
In[47]:= resitve = Solve[{ψ4[0] == 0, ψ4'[0] == 0, ψ4[L] == 1,
              ψ4''[L] == 0, ψ4'''[0] == 0}, neznanke];
         Do[M[i] = M[i] /. resitve[[1]], {i, 0, 4}];
```

in interpolacijska funkcija je tako:

In[49]:= **ψ4[x]**

Out[49]= $\frac{6\,x^2}{5\,L^2} - \frac{x^4}{5\,L^4}$

Njena slika je (narisana npr. v svetlo zeleni barvi):

```
In[50]:= funkcija4 = Plot[ψ4[x] /. L -> 1, {x, 0, 1}, PlotRange -> All,
           PlotStyle -> RGBColor[0.5, 0.8, 0.2]];
```



Tudi ta slika se bistveno ne razlikuje od prejšnjih in njej pripadajoči posplošeni dinamični parametri so:

In[51]:= $\mathbf{mz} = \int_0^L \mathbf{m}\,\psi 4[\mathbf{x}]^2\, d\mathbf{x}$

$\mathbf{kz} = \int_0^L \mathbf{EI}\,\psi 4''[\mathbf{x}]^2\, d\mathbf{x}$

$\mathbf{kg} = \int_0^L \psi 4'[\mathbf{x}]^2\, d\mathbf{x}$

Out[51]= $\frac{1763\,L\,m}{7875}$

Out[52]= $\frac{384\,EI}{125\,L^3}$

Out[53]= $\frac{1088}{875\,L}$

Glede na dejstvo, da so bili upoštevani robni pogoji, ki ustrezajo situaciji pri nastanku kritične uklonske sile, je smiselno pričakovati do sedaj najboljši približek kritične uklonske sile:

```
In[54]:= Solve[kz - osna kg == 0, osna]
```

$$\text{Out[54]=}\ \left\{\left\{\text{osna}\to\frac{42\,\text{EI}}{17\,\text{L}^2}\right\}\right\}$$

Za boljšo predstavo je potrebno pretvoriti še ulomek:

```
In[55]:= N[%]
```

$$\text{Out[55]=}\ \left\{\left\{\text{osna}\to\frac{2.47059\,\text{EI}}{\text{L}^2}\right\}\right\}$$

kar potrdi predpostavko, saj napaka glede na znano analitično rešitev znaša zgolj še 0.15 %.

Hkrati pa je mogoče tudi pričakovati, da dobljeni približek lastne frekvence ne bo najboljši izmed vseh do sedaj izračunanih:

$$\text{In[56]:=}\ \text{N}\left[\text{Sqrt}\left[\frac{\text{kz}}{\text{mz}}\right]\right]$$

$$\text{Out[56]=}\ 3.70433\sqrt{\frac{\text{EI}}{\text{L}^4\,\text{m}}}$$

kar se tudi izkaže kot korektno (napaka znaša še vedno sprejemljivih 5.4 %).

Pripadajoča vrednost posplošene obtežbe pa je:

$$\text{In[57]:=}\ \text{pz} = \int_0^{\text{L}} \psi 4[\text{x}]\,d\text{x}$$

$$\text{Out[57]=}\ \frac{9\,\text{L}}{25}$$

Do sedaj se je izkazalo, da uporaba polinomov višjega reda, seveda ob uporabi ustreznih robnih pogojev, vodi do kvalitetnejših rešitev, kar je seveda bilo popolnoma pričakovano, saj polinom višjega reda lahko predstavlja boljši približek korektni rešitvi diferencialne enačbe (ki pa jo dejansko predstavljajo trigonometrični izrazi).

Zato se kot naslednja funkcija uporabi še nepopolni polinom višjega reda in neznane konstante so tokrat npr. L[1], L[2],....

■ Peta funkcija - polinom visjega reda

```
In[58]:= polinom = 0;
         neznanke = Table[L[i], {i, 1, 4}];
```

Pri izbiri členov se npr. izpusti konstantni člen:

```
In[60]:= Do[polinom += L[i] x^i, {i, 1, 4}]
```

in tako sledi nepopolni polinom 4. reda:

In[61]:= **ψ5[x_] = polinom**

Out[61]= $x\,L[1] + x^2\,L[2] + x^3\,L[3] + x^4\,L[4]$

Za določitev neznanih konstant se uporabijo tisti pogoji, ki so do sedaj vodili v splošnem do najkvalitetnejših približkov obeh iskanih veličin (v drugi funkciji):

In[62]:= **resitve = Solve[{ψ5[0] == 0, ψ5′[0] == 0, ψ5[L] == 1, ψ5''[L] == 0}, neznanke]**

```
Solve::svars : Equations may not
  give solutions for all "solve" variables.
```

Out[62]= $\left\{\left\{L[2] \to \frac{3}{2\,L^2} + \frac{3}{2}\,L^2\,L[4],\ L[3] \to -\frac{1}{2\,L^3} - \frac{5}{2}\,L\,L[4],\ L[1] \to 0\right\}\right\}$

vendar *Mathematica* sedaj ne more poiskati enolične rešitve, kar napiše tudi v opozorilu, hkrati pa je to razvidno tudi iz rezultata, saj rešitev za L[2] vsebuje tudi L[4]. Razlog za probleme ni v *Mathematici*, ampak v slabo izbranem polinomu, saj odsotnost linearnega člena vodi do prvega pogoja v obliki 0=0, v katerem pa ne nastopajo neznane konstante in so tako v bistvu definirane zgolj tri enačbe s štirimi neznankami.
Pri izbiri nepopolnih polinomov je torej potrebna posebna previdnost.

Ker je znano, da dejanske rešitve diferencialne enačbe pomikov vsebujejo trigonometrične funkcije, se kot šesta interpolacijska funkcija uporabi rešitev, ki jo sestavljajo zgolj trigonometrične funkcije (čeprav je mogoče brez težav kombinirati tudi polinome in trigonometrične funkcije).
Pri izbiri členov *sinus* in *cosinus* je potrebno sedaj paziti, da se izbereta taka, ki zgolj naraščata proti prostemu koncu konzole, npr. takšna, ki na koncu intervala dosežeta svoj maksimum:

■ Sesta funkcija - trigonometricna funkcija

In[63]:= $\boldsymbol{\psi6[x\_] := c1\,Sin\left[\frac{\pi}{2}\,\frac{x}{L}\right] + c2\,Cos\left[\frac{\pi}{2}\,\frac{x}{L}\right] + c3\,Sinh\left[\frac{\pi}{2}\,\frac{x}{L}\right] + c4\,Cosh\left[\frac{\pi}{2}\,\frac{x}{L}\right];}$

Spet se uporabijo najuspešnejši robni pogoji:

In[64]:= **resitve = Solve[{ψ6[0] == 0, ψ6′[0] == 0, ψ6[L] == 1, ψ6''[L] == 0}, {c1, c2, c3, c4}]**

Out[64]= $\left\{\left\{c2 \to \frac{1}{2}\left(-Sech\left[\frac{\pi}{2}\right] - Tanh\left[\frac{\pi}{2}\right]\right),\ c4 \to \frac{1}{2}\left(Sech\left[\frac{\pi}{2}\right] + Tanh\left[\frac{\pi}{2}\right]\right),\ c1 \to \frac{1}{2},\ c3 \to -\frac{1}{2}\right\}\right\}$

ki tokrat vodijo do rešitev, ki pa se jih še poskusi nekoliko poenostaviti:

In[65]:= **Simplify[resitve]**

Out[65]= $\left\{\left\{c2 \to -\frac{1}{2}\,\mathrm{Sech}\left[\frac{\pi}{2}\right]\left(1+\mathrm{Sinh}\left[\frac{\pi}{2}\right]\right),\ c4 \to \frac{1}{2}\left(\mathrm{Sech}\left[\frac{\pi}{2}\right]+\mathrm{Tanh}\left[\frac{\pi}{2}\right]\right),\ c1 \to \frac{1}{2},\ c3 \to -\frac{1}{2}\right\}\right\}$

vendar brezuspešno. Zato se izračunane vrednosti kar dodeli konstantam:

```
In[66]:= c1=c1/.resitve[[1]];
         c2=c2/.resitve[[1]];
         c3=c3/.resitve[[1]];
         c4=c4/.resitve[[1]];
```

in funkcija dobi obliko:

In[70]:= **ψ6[x]**

Out[70]= $\frac{1}{2}\,\mathrm{Sin}\left[\frac{\pi x}{2\,L}\right]-\frac{1}{2}\,\mathrm{Sinh}\left[\frac{\pi x}{2\,L}\right]+\frac{1}{2}\,\mathrm{Cos}\left[\frac{\pi x}{2\,L}\right]\left(-\mathrm{Sech}\left[\frac{\pi}{2}\right]-\mathrm{Tanh}\left[\frac{\pi}{2}\right]\right)+\frac{1}{2}\,\mathrm{Cosh}\left[\frac{\pi x}{2\,L}\right]\left(\mathrm{Sech}\left[\frac{\pi}{2}\right]+\mathrm{Tanh}\left[\frac{\pi}{2}\right]\right)$

ki pa se lahko še nekoliko poenostavi:

In[71]:= **N[ψ6[x]]**

Out[71]= $-0.657845\,\mathrm{Cos}\left[\frac{1.5708\,x}{L}\right]+0.657845\,\mathrm{Cosh}\left[\frac{1.5708\,x}{L}\right]+0.5\,\mathrm{Sin}\left[\frac{1.5708\,x}{L}\right]-0.5\,\mathrm{Sinh}\left[\frac{1.5708\,x}{L}\right]$

Poenostavitev pa seveda ne vpliva na izris funkcije (npr. v vijolični barvi):

```
In[72]:= funkcija6 = Plot[ψ6[x] /. L -> 1, {x, 0, 1},
           PlotStyle -> RGBColor[0.5, 0, 1]];
```



Ker je slika funkcije očitno podobne prejšnjim, po že ustaljenem vzorcu sledi izračun posplošene mase, posplošene togosti ter posplošene geometrijske togosti, vendar očitno *Mathematica* naleti na nepričakovane težave pri izračunu posplošene mase, saj naleti na nedoločeni oz. nedefinirani člen, na kar opozori tudi z izpisom *Indeterminate*:

In[73]:= $mz = \int_0^L m\,\psi6[x]^2\,dx$

$kz = \int_0^L EI\,\psi6''[x]^2\,dx$

$kg = \int_0^L Pukl\,\psi6'[x]^2\,dx$

Out[73]= Indeterminate

Out[74]= $\dfrac{EI\,\pi^3\,\mathrm{Sech}\left[\frac{\pi}{2}\right]^2\left(4\,\mathrm{Cosh}\left[\frac{\pi}{2}\right]+2\,\pi\left(1+\mathrm{Sinh}\left[\frac{\pi}{2}\right]\right)^2\right)}{128\,L^3}$

Out[75]= $Pukl\left(-\dfrac{\pi\left(\mathrm{Sech}\left[\frac{\pi}{2}\right]+\mathrm{Tanh}\left[\frac{\pi}{2}\right]\right)}{8\,L}+\dfrac{\pi\left(\pi+6\,\mathrm{Tanh}\left[\frac{\pi}{2}\right]\right)}{16\,L}\right)$

Problem se poskusi rešiti z izračunom nedoločenega integrala:

```
In[76]:= mz = Integrate[m ψ6[x]^2, x]
```

$$
\text{Out[76]=}\ \Bigg(L\,m\left(\frac{\pi x}{L}+2\,\mathrm{Cos}\left[\frac{\pi x}{L}\right]\mathrm{Cosh}\left[\frac{\pi}{2}\right]+\frac{\pi x\,\mathrm{Cosh}[\pi]}{L}-2\,\mathrm{Cosh}\left[\frac{\pi}{2}-\frac{\pi x}{L}\right]+4\,\mathrm{Cos}\left[\frac{\pi x}{2L}\right]\mathrm{Cosh}\left[\frac{\pi}{2}-\frac{\pi x}{2L}\right]-4\,\mathrm{Cos}\left[\frac{\pi x}{2L}\right]\mathrm{Cosh}\left[\frac{\pi}{2}+\frac{\pi x}{2L}\right]-4\,\mathrm{Cosh}\left[\frac{\pi x}{2L}\right]\mathrm{Sin}\left[\frac{\pi x}{2L}\right]-4\,\mathrm{Cosh}\left[\pi-\frac{\pi x}{2L}\right]\mathrm{Sin}\left[\frac{\pi x}{2L}\right]+\frac{4\pi x\,\mathrm{Sinh}\left[\frac{\pi}{2}\right]}{L}+2\,\mathrm{Sin}\left[\frac{\pi x}{L}\right]\mathrm{Sinh}\left[\frac{\pi}{2}\right]+\mathrm{Cos}\left[\frac{\pi x}{L}\right]\mathrm{Sinh}[\pi]+\mathrm{Sinh}\left[\frac{\pi x}{L}\right]-\mathrm{Sinh}\left[\pi-\frac{\pi x}{L}\right]-8\,\mathrm{Sin}\left[\frac{\pi x}{2L}\right]\mathrm{Sinh}\left[\frac{\pi}{2}-\frac{\pi x}{2L}\right]\right)\left(\frac{1}{2}\mathrm{Sin}\left[\frac{\pi x}{2L}\right]-\frac{1}{2}\mathrm{Sinh}\left[\frac{\pi x}{2L}\right]+\frac{1}{2}\mathrm{Cos}\left[\frac{\pi x}{2L}\right]\left(-\mathrm{Sech}\left[\frac{\pi}{2}\right]-\mathrm{Tanh}\left[\frac{\pi}{2}\right]\right)+\frac{1}{2}\mathrm{Cosh}\left[\frac{\pi x}{2L}\right]\left(\mathrm{Sech}\left[\frac{\pi}{2}\right]+\mathrm{Tanh}\left[\frac{\pi}{2}\right]\right)\right)\text{^}2\Bigg)\Big/\left(2\pi\left(\mathrm{Cos}\left[\frac{\pi x}{2L}\right]-\mathrm{Cosh}\left[\frac{\pi x}{2L}\right]-\mathrm{Cosh}\left[\frac{\pi}{2}\right]\mathrm{Sin}\left[\frac{\pi x}{2L}\right]+\mathrm{Cos}\left[\frac{\pi x}{2L}\right]\mathrm{Sinh}\left[\frac{\pi}{2}\right]-\mathrm{Sinh}\left[\frac{\pi}{2}-\frac{\pi x}{2L}\right]\right)\text{^}2\right)
$$

ki vodi do izraza, ki pa se ga da še krajšati oz. poenostaviti:

In[77]:= **mz = Simplify[mz]**

Out[77]= $$\frac{1}{8\pi}\left(m\,\mathrm{Sech}\left[\frac{\pi}{2}\right]^2\left(\pi x+2L\,\mathrm{Cos}\left[\frac{\pi x}{L}\right]\mathrm{Cosh}\left[\frac{\pi}{2}\right]+\pi x\,\mathrm{Cosh}[\pi]-2L\,\mathrm{Cosh}\left[\frac{\pi(L-2x)}{2L}\right]+4L\,\mathrm{Cos}\left[\frac{\pi x}{2L}\right]\mathrm{Cosh}\left[\frac{\pi(L-x)}{2L}\right]-4L\,\mathrm{Cos}\left[\frac{\pi x}{2L}\right]\mathrm{Cosh}\left[\frac{\pi(L+x)}{2L}\right]-4L\,\mathrm{Cosh}\left[\frac{\pi x}{2L}\right]\mathrm{Sin}\left[\frac{\pi x}{2L}\right]-4L\,\mathrm{Cosh}\left[\pi-\frac{\pi x}{2L}\right]\mathrm{Sin}\left[\frac{\pi x}{2L}\right]+4\pi x\,\mathrm{Sinh}\left[\frac{\pi}{2}\right]+2L\,\mathrm{Sin}\left[\frac{\pi x}{L}\right]\mathrm{Sinh}\left[\frac{\pi}{2}\right]+L\,\mathrm{Cos}\left[\frac{\pi x}{L}\right]\mathrm{Sinh}[\pi]-8L\,\mathrm{Sin}\left[\frac{\pi x}{2L}\right]\mathrm{Sinh}\left[\frac{\pi(L-x)}{2L}\right]+L\,\mathrm{Sinh}\left[\frac{\pi x}{L}\right]-L\,\mathrm{Sinh}\left[\pi-\frac{\pi x}{L}\right]\right)\right)$$

Sedaj se lahko izračunata vrednost funkcije pri zgornji meji, torej pri x=L:

In[78]:= **mzL = mz /. x → L**

Out[78]= $$\frac{m\,\mathrm{Sech}\left[\frac{\pi}{2}\right]^2\left(L\pi-12L\,\mathrm{Cosh}\left[\frac{\pi}{2}\right]+L\pi\,\mathrm{Cosh}[\pi]+4L\pi\,\mathrm{Sinh}\left[\frac{\pi}{2}\right]\right)}{8\pi}$$

ter še pri spodnji meji (pri x=0):

In[79]:= **mz0 = mz /. x → 0**

Out[79]= 0

Tako je sedaj mogoče izračunati vrednost določenega integrala kot:

In[80]:= **mz = mzL - mz0**

Out[80]= $$\frac{m\,\mathrm{Sech}\left[\frac{\pi}{2}\right]^2\left(L\pi-12L\,\mathrm{Cosh}\left[\frac{\pi}{2}\right]+L\pi\,\mathrm{Cosh}[\pi]+4L\pi\,\mathrm{Sinh}\left[\frac{\pi}{2}\right]\right)}{8\pi}$$

Nato se lahko izračunajo še kritična uklonska sila, približek prve lastne frekvence ter posplošena obtežba:

```
In[81]:= N[Solve[kz - kg == 0, Pukl]]
         N[Sqrt[kz/mz]]
         pz = ∫_0^L ψ6[x] dx
```

Out[81]= $\left\{\left\{\text{Pukl} \to \frac{2.55856\,\text{EI}}{L^2}\right\}\right\}$

Out[82]= $3.52965\sqrt{\frac{\text{EI}}{L^4\,m}}$

Out[83]= Indeterminate

kjer pa spet nastopijo težave (nedoločenost) pri izračunu posplošene sile. Zato se ponovno izračuna nedoločeni integral:

```
In[84]:= pz = ∫ψ6[x] dx;
```

ki se poenostavi, da se lažje ugotovi vzrok nedoločenosti:

```
In[85]:= pz = Simplify[pz]
```

Out[85]=
$$-\left(L\,\mathrm{Sech}\left[\frac{\pi}{2}\right]\left(2\,\mathrm{Cos}\left[\frac{\pi x}{L}\right]\mathrm{Cosh}\left[\frac{\pi}{2}\right]-2\,\mathrm{Cosh}\left[\frac{\pi(L-2x)}{2L}\right]+2\,\mathrm{Cos}\left[\frac{\pi x}{2L}\right]\mathrm{Cosh}\left[\frac{\pi(L-x)}{2L}\right]-2\,\mathrm{Cos}\left[\frac{\pi x}{2L}\right]\mathrm{Cosh}\left[\frac{\pi(L+x)}{2L}\right]-2\,\mathrm{Cosh}\left[\frac{\pi x}{2L}\right]\mathrm{Sin}\left[\frac{\pi x}{2L}\right]-2\,\mathrm{Cosh}\left[\pi-\frac{\pi x}{2L}\right]\mathrm{Sin}\left[\frac{\pi x}{2L}\right]+2\,\mathrm{Sin}\left[\frac{\pi x}{L}\right]\mathrm{Sinh}\left[\frac{\pi}{2}\right]+\mathrm{Cos}\left[\frac{\pi x}{L}\right]\mathrm{Sinh}[\pi]-4\,\mathrm{Sin}\left[\frac{\pi x}{2L}\right]\mathrm{Sinh}\left[\frac{\pi(L-x)}{2L}\right]+\mathrm{Sinh}\left[\frac{\pi x}{L}\right]-\mathrm{Sinh}\left[\pi-\frac{\pi x}{L}\right]\right)\right)\Big/\left(2\pi\left(\mathrm{Cos}\left[\frac{\pi x}{2L}\right]-\mathrm{Cosh}\left[\frac{\pi x}{2L}\right]-\mathrm{Cosh}\left[\frac{\pi}{2}\right]\mathrm{Sin}\left[\frac{\pi x}{2L}\right]+\mathrm{Cos}\left[\frac{\pi x}{2L}\right]\mathrm{Sinh}\left[\frac{\pi}{2}\right]-\mathrm{Sinh}\left[\frac{\pi(L-x)}{2L}\right]\right)\right)$$

Vrednost funkcije pri zgornji meji je tako:

```
In[86]:= pzL = pz /. x → L
```

Out[86]= $-\frac{2\,L\,\mathrm{Sech}\left[\frac{\pi}{2}\right]}{\pi}$

Pri spodnji meji:

```
In[87]:= pz0 = pz /. x → 0

      Power::infy : Infinite expression 1/0 encountered.

Out[87]= Indeterminate
```

pa ponovno pride do nedoločenosti, saj *Mathematica* zazna deljenje z nič. Zato se najprej poišče števec izraza z ukazom *Numerator[izraz]:*

In[88]:= **stevec = Numerator[pz]**

Out[88]= $-L\,\mathrm{Sech}\left[\frac{\pi}{2}\right]\left(2\,\mathrm{Cos}\left[\frac{\pi x}{L}\right]\mathrm{Cosh}\left[\frac{\pi}{2}\right]-2\,\mathrm{Cosh}\left[\frac{\pi(L-2x)}{2L}\right]+2\,\mathrm{Cos}\left[\frac{\pi x}{2L}\right]\mathrm{Cosh}\left[\frac{\pi(L-x)}{2L}\right]-2\,\mathrm{Cos}\left[\frac{\pi x}{2L}\right]\mathrm{Cosh}\left[\frac{\pi(L+x)}{2L}\right]-2\,\mathrm{Cosh}\left[\frac{\pi x}{2L}\right]\mathrm{Sin}\left[\frac{\pi x}{2L}\right]-2\,\mathrm{Cosh}\left[\pi-\frac{\pi x}{2L}\right]\mathrm{Sin}\left[\frac{\pi x}{2L}\right]+2\,\mathrm{Sin}\left[\frac{\pi x}{L}\right]\mathrm{Sinh}\left[\frac{\pi}{2}\right]+\mathrm{Cos}\left[\frac{\pi x}{L}\right]\mathrm{Sinh}[\pi]-4\,\mathrm{Sin}\left[\frac{\pi x}{2L}\right]\mathrm{Sinh}\left[\frac{\pi(L-x)}{2L}\right]+\mathrm{Sinh}\left[\frac{\pi x}{L}\right]-\mathrm{Sinh}\left[\pi-\frac{\pi x}{L}\right]\right)$

katerega vrednost pri zgornji meji je:

```
In[89]:= stevec /. x → 0

Out[89]= 0
```

Nato se poišče še imenovalec izraza z ukazom *Denominator[izraz]*:

In[90]:= **imenovalec = Denominator[pz]**

Out[90]= $2\pi\left(\mathrm{Cos}\left[\frac{\pi x}{2L}\right]-\mathrm{Cosh}\left[\frac{\pi x}{2L}\right]-\mathrm{Cosh}\left[\frac{\pi}{2}\right]\mathrm{Sin}\left[\frac{\pi x}{2L}\right]+\mathrm{Cos}\left[\frac{\pi x}{2L}\right]\mathrm{Sinh}\left[\frac{\pi}{2}\right]-\mathrm{Sinh}\left[\frac{\pi(L-x)}{2L}\right]\right)$

ki je prav tako enak nič pri spodnji meji:

```
In[91]:= imenovalec /. x → 0

Out[91]= 0
```

Očitno je, da gre za nedoločenost tipa $\frac{0}{0}$ in zato se uporabi l'Hospitalovo pravilo. Odvod števca:

```
In[92]:= stevec = D[stevec, x]
```

Out[92]=
$$-L\,\mathrm{Sech}\left[\frac{\pi}{2}\right]\left(-\frac{\pi\,\mathrm{Cos}\left[\frac{\pi x}{2L}\right]\mathrm{Cosh}\left[\frac{\pi x}{2L}\right]}{L}+\frac{\pi\,\mathrm{Cosh}\left[\frac{\pi x}{L}\right]}{L}+\frac{\pi\,\mathrm{Cosh}\left[\pi-\frac{\pi x}{L}\right]}{L}-\frac{\pi\,\mathrm{Cos}\left[\frac{\pi x}{2L}\right]\mathrm{Cosh}\left[\pi-\frac{\pi x}{2L}\right]}{L}+\frac{\pi\,\mathrm{Cosh}\left[\frac{\pi(L-x)}{2L}\right]\mathrm{Sin}\left[\frac{\pi x}{2L}\right]}{L}+\frac{\pi\,\mathrm{Cosh}\left[\frac{\pi(L+x)}{2L}\right]\mathrm{Sin}\left[\frac{\pi x}{2L}\right]}{L}-\frac{2\pi\,\mathrm{Cosh}\left[\frac{\pi}{2}\right]\mathrm{Sin}\left[\frac{\pi x}{L}\right]}{L}+\frac{2\pi\,\mathrm{Cos}\left[\frac{\pi x}{L}\right]\mathrm{Sinh}\left[\frac{\pi}{2}\right]}{L}-\frac{\pi\,\mathrm{Sin}\left[\frac{\pi x}{L}\right]\mathrm{Sinh}[\pi]}{L}+\frac{2\pi\,\mathrm{Sinh}\left[\frac{\pi(L-2x)}{2L}\right]}{L}-\frac{3\pi\,\mathrm{Cos}\left[\frac{\pi x}{2L}\right]\mathrm{Sinh}\left[\frac{\pi(L-x)}{2L}\right]}{L}-\frac{\pi\,\mathrm{Sin}\left[\frac{\pi x}{2L}\right]\mathrm{Sinh}\left[\frac{\pi x}{2L}\right]}{L}-\frac{\pi\,\mathrm{Cos}\left[\frac{\pi x}{2L}\right]\mathrm{Sinh}\left[\frac{\pi(L+x)}{2L}\right]}{L}+\frac{\pi\,\mathrm{Sin}\left[\frac{\pi x}{2L}\right]\mathrm{Sinh}\left[\pi-\frac{\pi x}{2L}\right]}{L}\right)$$

je na spodnji meji:

```
In[93]:= stevec /. x → 0
```

Out[93]= 0

še vedno enak nič. Poišče se še odvod imenovalca:

```
In[94]:= imenovalec = D[imenovalec, x]
```

Out[94]=
$$2\pi\left(-\frac{\pi\,\mathrm{Cos}\left[\frac{\pi x}{2L}\right]\mathrm{Cosh}\left[\frac{\pi}{2}\right]}{2L}+\frac{\pi\,\mathrm{Cosh}\left[\frac{\pi(L-x)}{2L}\right]}{2L}-\frac{\pi\,\mathrm{Sin}\left[\frac{\pi x}{2L}\right]}{2L}-\frac{\pi\,\mathrm{Sin}\left[\frac{\pi x}{2L}\right]\mathrm{Sinh}\left[\frac{\pi}{2}\right]}{2L}-\frac{\pi\,\mathrm{Sinh}\left[\frac{\pi x}{2L}\right]}{2L}\right)$$

ki pa je prav tako enak nič:

```
In[95]:= imenovalec /. x → 0
```

Out[95]= 0

Ker gre ponovno za nedoločenost enakega tipa, se še enkrat uporabi l'Hospitalovo pravilo. Izračuna se ponovni odvod števca:

```
In[96]:= stevec = D[stevec, x];
```

Izpis ni zanimiv, saj je pomembna zgolj diskretna vrednost pri x=0, ki pa tokrat ni enaka nič:

```
In[97]:= stevec /. x → 0
```

$$\text{Out[97]= } -L\,\mathrm{Sech}\left[\frac{\pi}{2}\right]\left(-\frac{2\,\pi^2\,\mathrm{Cosh}\left[\frac{\pi}{2}\right]}{L^2}-\frac{\pi^2\,\mathrm{Sinh}[\pi]}{L^2}\right)$$

Poiskati je potrebno še odvod imenovalca:

```
In[98]:= imenovalec = D[imenovalec, x];
```

Tudi ta izpis ni zanimiv, saj je pomembna zgolj diskretna vrednost pri x=0. Tudi ta je od nič različna:

```
In[99]:= imenovalec /. x → 0
```

$$\text{Out[99]= } 2\,\pi\left(-\frac{\pi^2}{2\,L^2}-\frac{\pi^2\,\mathrm{Sinh}\left[\frac{\pi}{2}\right]}{2\,L^2}\right)$$

kar omogoči izračun vrednosti integrala tudi pri spodnji meji:

```
In[100]:= stevec / imenovalec /. x → 0
```

$$\text{Out[100]= } -\frac{L\,\mathrm{Sech}\left[\frac{\pi}{2}\right]\left(-\frac{2\,\pi^2\,\mathrm{Cosh}\left[\frac{\pi}{2}\right]}{L^2}-\frac{\pi^2\,\mathrm{Sinh}[\pi]}{L^2}\right)}{2\,\pi\left(-\frac{\pi^2}{2\,L^2}-\frac{\pi^2\,\mathrm{Sinh}\left[\frac{\pi}{2}\right]}{2\,L^2}\right)}$$

Večino členov zgornjega izraza je mogoče kar izvrednotiti:

```
In[101]:= N[%]

Out[101]= -0.63662 L
```

Rezultat se lahko shrani:

```
In[102]:= pz0 = %

Out[102]= -0.63662 L
```

kar (končno) vodi do posplošene obtežbe:

```
In[103]:= pz = N[pzL - pz0]

Out[103]= 0.382903 L
```

Čeprav je sedanja izbira (trigonometrične interpolacijske) funkcije že jasno pokazala slabosti takšne odločitve (kljub največ vloženega dela ne sledijo najboljši rezultati), je potrebno opozoriti še na drugo slabost izbire trigonometričnih funkcij. Medtem kot pri polinomih jasno, kateri red polinoma izbrati (število uporabljenih robnih pogojev in morebitnih pogojev zveznosti jasno definira red polinoma), je pri istem številu informacij pri trigonometričnih interpolacijskih funkcijah mogoče izbrati različne člene, kar vodi do različno uspešnih funkcij.

Medtem ko so se členi šeste interpolacijske funkcije smiselno izbrali tako, da so naraščali proti prostemu koncu konzole, se na to dejstvo pri izbiri sedme interpolacijske funkcije ne bo več oziralo, kar bo pričakovano vodilo do najslabših rezultatov. Ena izmed možnih funkcij je tako npr.:

■ Sedma funkcija - trigonometricna funkcija

In[104]:=
```
ψ7[x_] := d1 Sin[π x/L] + d2 Cos[π x/L] + d3 Sinh[π x/L] +
    d4 Cosh[π x/L];
```

Po že ustaljenem postopku se zanjo poiščejo vrednosti konstant:

In[105]:=
```
resitve = Solve[{ψ7[0] == 0, ψ7'[0] == 0, ψ7[L] == 1,
    ψ7''[L] == 0}, {d1, d2, d3, d4}]
```

Out[105]= $\left\{\left\{d1\to\frac{1}{2}\,(\mathrm{Coth}[\pi]-\mathrm{Csch}[\pi]),\ d3\to\frac{1}{2}\,(-\mathrm{Coth}[\pi]+\mathrm{Csch}[\pi]),\ d2\to-\frac{1}{2},\ d4\to\frac{1}{2}\right\}\right\}$

ki se najprej poenostavijo:

In[106]:=
```
resitve = Simplify[resitve]
```

Out[106]= $\left\{\left\{d1\to\frac{1}{2}\,\mathrm{Tanh}\left[\frac{\pi}{2}\right],\ d3\to-\frac{1}{2}\,\mathrm{Tanh}\left[\frac{\pi}{2}\right],\ d2\to-\frac{1}{2},\ d4\to\frac{1}{2}\right\}\right\}$

nato pa dodelijo samim konstantam:

In[107]:=
```
d1=d1/.resitve[[1]];
d2=d2/.resitve[[1]];
d3=d3/.resitve[[1]];
d4=d4/.resitve[[1]];
```

Funkcija je tako:

In[111]:=
```
ψ7[x]
```

Out[111]= $-\frac{1}{2}\,\mathrm{Cos}\left[\frac{\pi x}{L}\right]+\frac{1}{2}\,\mathrm{Cosh}\left[\frac{\pi x}{L}\right]+\frac{1}{2}\,\mathrm{Sin}\left[\frac{\pi x}{L}\right]\,\mathrm{Tanh}\left[\frac{\pi}{2}\right]-\frac{1}{2}\,\mathrm{Sinh}\left[\frac{\pi x}{L}\right]\,\mathrm{Tanh}\left[\frac{\pi}{2}\right]$

Nekatere člene je mogoče še izvrednotiti:

```
In[112]:= N[ψ7[x]]
```

$$\text{Out[112]=}\ -0.5\,\mathrm{Cos}\left[\frac{3.14159\,x}{L}\right]+0.5\,\mathrm{Cosh}\left[\frac{3.14159\,x}{L}\right]+0.458576\,\mathrm{Sin}\left[\frac{3.14159\,x}{L}\right]-0.458576\,\mathrm{Sinh}\left[\frac{3.14159\,x}{L}\right]$$

Pravo predstavo o funkciji pa seveda nudi njen izris (npr. v oranžni barvi):

```
In[113]:= funkcija7 = Plot[ψ7[x] /. L -> 1, {x, 0, 1},
            PlotStyle -> RGBColor[1, 0.5, 0]];
```



Čeprav je mogoče že kar na osnovi slike (inženirski pristop) soditi o (pričakovani ne)primernosti funkcije, se vseeno poskusijo (matematični dokaz) izračunati koeficienti, ki omogočajo izračun lastne frekvence in kritične uklonske sile, ki pa predstavljajo popolnoma objektivno analizo primernosti ali neprimernosti posamezne funkcije.

Izpis posameznega člena sicer ni zanimiv, je pa nujno potreben, saj v primeru uporabe podpičja na koncu ukaza *Mathematica* ne bo izpisala niti opozorila o morebitni nedoločenosti izraza, kar bo onemogočilo zaznavanje napake oz. težav pri izračunu. Tako bi npr. težavi pri izračunu posplošene mase in poslošene togosti ostale skriti vse do izračuna približka lastne frekvence oz. kritične uklonske sile.

In[114]:= $\mathbf{mz} = \int_0^L \mathbf{m}\,\psi\mathbf{7[x]}^2\,d\mathbf{x}$

$\mathbf{kz} = \int_0^L \mathbf{EI}\,\psi\mathbf{7''[x]}^2\,d\mathbf{x}$

$\mathbf{kg} = \int_0^L \mathbf{Pukl}\,\psi\mathbf{7'[x]}^2\,d\mathbf{x}$

Out[114]= $\Big(m\,\Big(60\,L\,\pi + 60\,L\,\pi\,\mathrm{Cosh}[\pi] + 24\,L\,\pi\,\mathrm{Cosh}[2\,\pi] + 4\,L\,\pi\,\mathrm{Cosh}[3\,\pi] - 4\,L\,\pi\,\mathrm{Cosh}\left[\frac{3\,\pi}{2}\right]\mathrm{Sech}\left[\frac{\pi}{2}\right] - 12\,L\,\pi\,\mathrm{Cosh}\left[\frac{5\,\pi}{2}\right]\mathrm{Sech}\left[\frac{\pi}{2}\right] - 4\,L\,\pi\,\mathrm{Cosh}\left[\frac{7\,\pi}{2}\right]\mathrm{Sech}\left[\frac{\pi}{2}\right] + 3\,L\,\pi\,\mathrm{Sech}\left[\frac{\pi}{2}\right]^2 - 4\,L\,\pi\,\mathrm{Cosh}[2\,\pi]\,\mathrm{Sech}\left[\frac{\pi}{2}\right]^2 + L\,\pi\,\mathrm{Cosh}[4\,\pi]\,\mathrm{Sech}\left[\frac{\pi}{2}\right]^2 + 60\,L\,\mathrm{Sinh}[\pi] + 18\,L\,\mathrm{Sech}\left[\frac{\pi}{2}\right]^2\mathrm{Sinh}[\pi] + 36\,L\,\mathrm{Sech}\left[\frac{\pi}{2}\right]\mathrm{Sinh}\left[\frac{3\,\pi}{2}\right] + 48\,L\,\mathrm{Sinh}[2\,\pi] - 6\,L\,\mathrm{Sech}\left[\frac{\pi}{2}\right]^2\mathrm{Sinh}[2\,\pi] - 12\,L\,\mathrm{Sech}\left[\frac{\pi}{2}\right]\mathrm{Sinh}\left[\frac{5\,\pi}{2}\right] + 12\,L\,\mathrm{Sinh}[3\,\pi] - 6\,L\,\mathrm{Sech}\left[\frac{\pi}{2}\right]^2\mathrm{Sinh}[3\,\pi] - 12\,L\,\mathrm{Sech}\left[\frac{\pi}{2}\right]\mathrm{Sinh}\left[\frac{7\,\pi}{2}\right] + 3\,L\,\mathrm{Sech}\left[\frac{\pi}{2}\right]^2\mathrm{Sinh}[4\,\pi] + 36\,L\,\mathrm{Tanh}\left[\frac{\pi}{2}\right]\Big)\Big) \Big/ (256\,\pi\,(1 + \mathrm{Cosh}[\pi]))$

Out[115]= `ComplexInfinity`

Out[116]= `Indeterminate`

Za izračun posplošene togosti se sedaj izvede izračun nedoločenega integrala:

In[117]:= $\mathbf{kz = Integrate[m\,\psi 7''[x]^2,\ x];}$

ki se še najprej uredi:

In[118]:= **kz = Simplify[kz]**

$$\text{Out[118]=}\ \frac{1}{16\,L^4}\left(m\,\pi^3\,\text{Sech}\left[\frac{\pi}{2}\right]^2\left(2\,\pi\,x+2\,\pi\,x\,\text{Cosh}[\pi]+4\,L\,\text{Cosh}\left[\pi-\frac{\pi\,x}{L}\right]\text{Sin}\left[\frac{\pi\,x}{L}\right]+L\,\text{Sin}\left[\frac{2\,\pi\,x}{L}\right]+L\,\text{Cos}\left[\frac{2\,\pi\,x}{L}\right]\text{Sinh}[\pi]+4\,L\,\text{Cos}\left[\frac{\pi\,x}{L}\right]\text{Sinh}\left[\frac{\pi\,x}{L}\right]-L\,\text{Sinh}\left[\pi-\frac{2\,\pi\,x}{L}\right]\right)\right)$$

Sedaj se lahko izračunata njegova vrednost pri zgornji meji, torej pri x=L:

In[119]:= **kzL = kz /. x → L**

$$\text{Out[119]=}\ \frac{m\,\pi^3\,\text{Sech}\left[\frac{\pi}{2}\right]^2\,(2\,L\,\pi+2\,L\,\pi\,\text{Cosh}[\pi]-2\,L\,\text{Sinh}[\pi])}{16\,L^4}$$

ter še pri spodnji meji (pri x=0):

In[120]:= **kz0 = kz /. x → 0**

Out[120]= 0

Ker je bil sedaj izračun vrednosti integrala pri obeh mejah izveden brez numeričnih težav, je mogoče izračunati tudi vrednost določenega integrala, ki predstavlja vrednost posplošene togosti:

In[121]:= **kz = (kzL - kz0) EI**

$$\text{Out[121]=}\ \frac{EI\,m\,\pi^3\,\text{Sech}\left[\frac{\pi}{2}\right]^2\,(2\,L\,\pi+2\,L\,\pi\,\text{Cosh}[\pi]-2\,L\,\text{Sinh}[\pi])}{16\,L^4}$$

Tako je mogoče sedaj izračunati vrednost približka prve lastne frekvence:

$$\text{In[122]:=}\ \mathbf{N\left[Sqrt\left[\frac{kz}{mz}\right]\right]}$$

$$\text{Out[122]=}\ 6.06373\,\sqrt{\frac{EI}{L^4}}$$

ki pa je, kot je bilo že pričakovano na osnovi izrisa slike, daleč najslabši. Zaradi tega tudi ni smiselno iskati približka vrednosti kritične uklonske sile.

Končna primerjava vseh do sedaj analiziranih funkcij pa pokaže, da so si vse pravzaprav medsebojno precej podobne (seveda z izjemo zadnje, sedme):

```
In[123]:= Show[funkcija1, funkcija2, funkcija3, funkcija4,
            funkcija6, funkcija7];
```



Ker pa je bila v prejšnjem podpoglavju analitično (na osnovi numerično izračunanih lastnih frekvenc) izračunana tudi nihajna oblika, ki ustreza rešitvi diferencialne enačbe, je smiselno to rešitev prenesti v sedanjo analizo in izvesti vsaj grafično primerjavo. Tako sledi (po prenosu in pretvorbi izhodne vrstice v vhodno):

■ Resitev diferencialne enacbe

```
In[124]:= ψ8[x_]:=C[1]*Cos[(1.8751040687119611*x)/L] -
          C[1]*Cosh[(1.8751040687119611*x)/L] -
          0.7340955137589127*C[1]*Sin[(1.8751040687119611*x)/
             L] + 0.7340955137589127*C[1]*
           Sinh[(1.8751040687119611*x)/L]
```

Ki se lahko izriše (ker sedma interpolacijska funkcija ni več zanima, se tudi za osmo funkcijo npr. spet uporabi dobro vidna oranžna barva), kot vrednost konstante C[1] pa se upošteva npr. vrednost 1:

```
In[125]:= funkcija8 = Plot[ψ8[x] /. {L -> 1, C[1] → 1}, {x, 0, 1},
            PlotStyle -> RGBColor[1, 0.5, 0]];
```



Izris pokaže na pomembno dejstvo (ki bi ga lahko opazili že pri izrisu funkcije v datoteki, kjer je bila izpeljana), namreč da vrednost funkcije na prostem koncu ni enaka nič, temveč –2, kar zahteva ponoven izris, tokrat z ustrezno korigirano vrednostjo koeficienta C[1]:

```
In[126]:= funkcija8 = Plot[ψ8[x] /. {L -> 1, C[1] → -0.5},
            {x, 0, 1}, PlotStyle -> RGBColor[1, 0.5, 0]];
```



in šele sedaj je smiselno izrisati in primerjati vse (uporabne) interpolacijske funkcije ter dejansko rešitev diferencialne enačbe:

```
In[127]:= Show[funkcija1, funkcija2, funkcija3, funkcija4,
           funkcija6, funkcija8];
```



ki ponovno potrdi dejstvo, ki je dejansko potrjeno že tudi z numeričnimi rezultati, da se lahko z kvalitetno izbiro interpolacijskih funkcij doseže kvaliteten približek dejanske nihajne oblike (amplituda ni pomembna, zgolj oblika).